

Survival Models in SAS - Part 2: PROC LIFETEST


November 21, 2007

Charlie Hallahan

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST

These talks are based on the book “**Survival Analysis Using the SAS System: A Practical Guide**” (1995) by Paul Allison.

The book is part of the SAS Books-by-Users series and can be found at <http://www.sas.com/apps/pubscat/bookdetails.jsp?catid=1&pc=55233>



Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST

This series of talks will cover

Chapter 1: Introduction

Chapter 2: Basic Concepts of Survival Analysis

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG

Chapter 5: Estimating Cox Regression Models with PROC PHREG

Chapter 6: Competing Risks

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST

Topics in Chapter 3:

Introduction

The Kaplan-Meier Method

Testing for Differences in Survivor Functions

The Life-Table Method

Life-Tables from Grouped Data

Testing for Effects of Covariates

Log Survivor and Smoothed Hazard Plots

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: Testing for Differences in Survivor Functions

In situations where a comparison is made between a standard and experimental treatment, it is natural to ask “*Did the experimental treatment make a difference in the survival experience of the two groups?*”

To test: $H_0: S_1(t) = S_2(t)$ for all t

PROC LIFETEST calculates the following tests:

1. **log-rank** or **Mantel-Haenszel** test
2. **Wilcoxon** test
3. **likelihood ratio test** (under the assumption that the event times have an exponential distribution).

For the `myelomatosis` data, we have the variable `treat` indicating observations using an experimental treatment.

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: Testing for Differences in Survivor Functions

Given two subgroups of the sample, let r = the number of unique event times for the two groups combined, d_{1j} = the number of deaths (events) that occur in group 1 at time j , and e_{1j} = the expected number of events in group 1 at time j .

As with 2x2 tables, $e_{1j} = n_{1j} \cdot d_j / n_j$ where n_j = total number of cases at risk just prior to time j , d_j = total number of events at time j in both groups, and n_{1j} = number at risk just prior to time j in group 1.

The **Wilcoxon** and **log - rank** tests are both based on the deviations $d_{1j} - e_{1j}$.

log - rank statistic:
$$\sum_{j=1}^r (d_{1j} - e_{1j})$$

Wilcoxon statistic:
$$\sum_{j=1}^r n_j (d_{1j} - e_{1j})$$

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: Testing for Differences in Survivor Functions

The associated **chi - square statistics** are obtained by squaring each statistic and dividing by its estimated variance.

Note that the Wilcoxon, as a weighted sum, gives more weight to the early times since n_j always decreases.

The **log - rank** test is more powerful for detecting differences of the form

$$S_1(t) = [S_2(t)]^\gamma \quad \text{for some } \gamma > 1. \quad (\text{i.e., a } \textit{proportional hazards model})$$

The **Wilcoxon** test is more powerful than the log-rank test in situations where event times have log-normal distributions with a common variance but with different means in the two groups.

Neither test is particularly good at detecting differences when survival curves cross.

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: Testing for Differences in Survivor Functions

```
title "Survival Curves for Two Treatment Groups";  
proc lifetest data=survival.myel plots=(s) graphics;  
    time dur*status(0);  
    strata treat;  
    symbol1 v=none color=black line=1;  
    symbol2 v=none color=black line=2;  
run;
```

The `strata` statement identifies the groups to be compared.

Note: a **likelihood-ratio test** is also reported. However, this test is usually inferior to the Wilcoxon and log-rank tests because the LR test requires the typically implausible assumption that the hazard function is constant in each group, implying an exponential distribution for event times.

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: Testing for Differences in Survivor Functions

Testing Homogeneity of Survival Curves for dur over Strata

| treat | Rank Statistics | |
|-------|-----------------|----------|
| | Log-Rank | Wilcoxon |
| 1 | -2.3376 | -18.000 |
| 2 | 2.3376 | 18.000 |

Covariance Matrix for the Log-Rank Statistics

| treat | 1 | 2 |
|-------|----------|----------|
| 1 | 4.16301 | -4.16301 |
| 2 | -4.16301 | 4.16301 |

Covariance Matrix for the Wilcoxon Statistics

| treat | 1 | 2 |
|-------|----------|----------|
| 1 | 1301.00 | -1301.00 |
| 2 | -1301.00 | 1301.00 |

Test of Equality over Strata


| Test | Chi-Square | DF | Pr > Chi-Square |
|-----------------|---------------|----------|-----------------|
| Log-Rank | 1.3126 | 1 | 0.2519 |
| Wilcoxon | 0.2490 | 1 | 0.6178 |
| -2Log(LR) | 1.5240 | 1 | 0.2170 |

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: Testing for Differences in Survivor Functions

The p-values on the previous page indicate no difference between the two survival curves.

They only really differ because there are no additional events in group 1 after 220 days and, thus, the curve for group 1 becomes constant after 220 days.

Note the much higher p-value for the **Wilcoxon** test since it places more weight on the earlier event times where the two curves are very close.



Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: Testing for Differences in Survivor Functions

Stratifying on the variable `renal` shows two survival curves that are clearly different.

Testing Homogeneity of Survival Curves for `dur` over Strata

| renal | Rank Statistics | |
|-------|-----------------|----------|
| | Log-Rank | Wilcoxon |
| 0 | -5.4009 | -113.00 |
| 1 | 5.4009 | 113.00 |

Covariance Matrix for the Log-Rank Statistics

| renal | 0 | 1 |
|-------|----------|----------|
| 0 | 1.21443 | -1.21443 |
| 1 | -1.21443 | 1.21443 |


Covariance Matrix for the Wilcoxon Statistics

| renal | 0 | 1 |
|-------|----------|----------|
| 0 | 611.722 | -611.722 |
| 1 | -611.722 | 611.722 |

Test of Equality over Strata

| Test | Chi-Square | DF | Pr > |
|-----------------|----------------|----------|------------------|
| | | | Chi-Square |
| Log-Rank | 24.0188 | 1 | <.0001 |
| Wilcoxon | 20.8739 | 1 | <.0001 |
| -2Log(LR) | 35.4813 | 1 | <.0001 |

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: Testing for Differences in Survivor Functions



Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: Testing for Differences in Survivor Functions

The `strata` statement is flexible enough to compare three or more survival curves.

For example, specifying: `strata treat renal;`

would define four separate groups for the four distinct cross-classified values.

In some situations, the number of distinct groups can grow quite large. In this case, the user can specify how the groups should be defined.

For example, using the variable `age`, the statement

```
strata age (20 30 40 50);
```

would produce five groups corresponding to the intervals:

$(-\infty, 20)$, $[20, 30)$, $[30, 40)$, $[40, 50)$, $[50, \infty)$

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: The Life-Table Method

For large datasets with many distinct event times, the **KM** method is not practical.

The **life-table** (or **actuarial**) **method** is designed for such datasets.

The range of event times is divided into intervals, either by a default method or by user-specified specifications.

While this grouping of the data allows for analysis of large datasets, it is also a drawback to the method because of its *arbitrariness*.

PROC LIFETEST, however, will calculate the **log-rank** and **Wilcoxon tests** using the *ungrouped* data.

The dataset used to illustrate this method, `recid`, consists of observations on 432 male inmates who were released from Maryland state prisons in the early 1970s.

They were followed for one year after release and the number of weeks to their first arrest was recorded.

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: The Life-Table Method

RECID Dataset (1st 20 observations)

| Obs | week | age | arrest |
|-----|------|-----|--------|
| 1 | 20 | 27 | 1 |
| 2 | 17 | 18 | 1 |
| 3 | 25 | 19 | 1 |
| 4 | 52 | 23 | 0 |
| 5 | 52 | 19 | 0 |
| 6 | 52 | 24 | 0 |
| 7 | 23 | 25 | 1 |
| 8 | 52 | 21 | 0 |
| 9 | 52 | 22 | 0 |
| 10 | 52 | 20 | 0 |
| 11 | 52 | 26 | 0 |
| 12 | 52 | 40 | 0 |
| 13 | 37 | 17 | 1 |
| 14 | 52 | 37 | 0 |
| 15 | 25 | 20 | 1 |
| 16 | 46 | 22 | 1 |
| 17 | 28 | 19 | 1 |
| 18 | 52 | 20 | 0 |
| 19 | 52 | 25 | 0 |
| 20 | 52 | 24 | 0 |

week = the week of 1st arrest (event time)

arrest = 1 if arrested during the year

i.e., arrest = 0 indicates a censored obs.

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: The Life-Table Method

```

title "Lifetable method estimate of the survival curve";
proc lifetest data=survival.recid method=life;
    time week*arrest(0);
run;

```

In this example, the default number of intervals is calculated to be six, each with a width of 10.

The user can specify their own intervals with either the **WIDTH = *w*** option or the **INTERVALS = *a,b,c,...*** option. The intervals can be of different lengths. The later intervals should be long enough to contain enough events to use in estimation.

Life Table Survival Estimates

| Interval [Lower, Upper) | Number Failed | Number Censored | Effective Sample Size | Conditional Probability of Failure | Conditional Probability Standard Error | Survival | Failure | Survival Standard Error |
|----------------------------|------------------|--------------------|-----------------------------|--|---|----------|---------|-------------------------------|
| 0 | 14 | 0 | 432.0 | 0.0324 | 0.00852 | 1.0000 | 0 | 0 |
| 10 | 21 | 0 | 418.0 | 0.0502 | 0.0107 | 0.9676 | 0.0324 | 0.00852 |
| 20 | 23 | 0 | 397.0 | 0.0579 | 0.0117 | 0.9190 | 0.0810 | 0.0131 |
| 30 | 23 | 0 | 374.0 | 0.0615 | 0.0124 | 0.8657 | 0.1343 | 0.0164 |
| 40 | 26 | 0 | 351.0 | 0.0741 | 0.0140 | 0.8125 | 0.1875 | 0.0188 |
| 50 | 7 | 318 | 166.0 | 0.0422 | 0.0156 | 0.7523 | 0.2477 | 0.0208 |

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: The Life-Table Method

Evaluated at the Midpoint of the Interval

| Interval [Lower, Upper) | | Median Residual Lifetime | Median Standard Error | PDF | PDF Standard Error | Hazard | Hazard Standard Error |
|----------------------------|----|--------------------------------|-----------------------------|---------|--------------------------|----------|-----------------------------|
| 0 | 10 | . | . | 0.00324 | 0.000852 | 0.003294 | 0.00088 |
| 10 | 20 | . | . | 0.00486 | 0.00103 | 0.005153 | 0.001124 |
| 20 | 30 | . | . | 0.00532 | 0.00108 | 0.005966 | 0.001244 |
| 30 | 40 | . | . | 0.00532 | 0.00108 | 0.006345 | 0.001322 |
| 40 | 50 | . | . | 0.00602 | 0.00114 | 0.007692 | 0.001507 |
| 50 | 60 | . | . | 0.00317 | 0.00118 | 0.004308 | 0.001628 |

Summary of the Number of Censored and Uncensored Values

| Total | Failed | Censored | Percent Censored |
|-------|--------|----------|---------------------|
| 432 | 114 | 318 | 73.61 |

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: The Life-Table Method

To explain the output, let the time intervals be defined as $[t_{i-1}, t_i)$, $i = 1, 2, \dots, k + 1$, where $t_0 = 0$ and $t_{k+1} = \infty$.

Let $n_i = \#$ of units entering the interval and let $d_i = \#$ of events (arrests) occurring in the interval.

Let $b_i = t_i - t_{i-1}$ and let $n'_i = n_i - w_i / 2$, where $w_i = \#$ of units censored in the interval.

The *effective sample size* of the interval $[t_{i-1}, t_i)$ is denoted by n'_i .

In our example, the only censored cases occur in the final interval, so $n'_i = n_i$ for $i \leq k$.

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: The Life-Table Method

The **life - table method** treats any cases censored within an interval as if they were censored *at the midpoint of the interval*. This assumes that the distribution of censoring times is uniform within the interval. Since censored cases are only at risk for half of the interval, they only count for half in figuring the effective sample size.

So the *effective sample size* for the last interval is $7+318/2 = 166$, where the seven men who were arrested are treated as though they were at risk for the whole interval.

The *Conditional Probability of Failure* is an estimate of the probability that an event will occur in the interval, given that no event has yet occurred, $\hat{q}_i = \frac{d_i}{n'_i}$, and $\hat{\sigma}_i(\hat{q}_i) = \sqrt{\frac{\hat{q}_i \hat{p}_i}{n'}}$, where $\hat{p}_i = 1 - \hat{q}_i$.

The estimate of the *survival function* is $\hat{S}(t_i) = \prod_{j=1}^{i-1} \hat{p}_{i-1} = \begin{cases} 1 & i = 0 \\ \hat{S}(t_{i-1}) \hat{p}_{i-1} & i > 0 \end{cases}$

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: The Life-Table Method

The major differences between the **KM** and **life-table methods** are:


- The number of censored observations in an interval is not halved in the KM estimator.
- The interval boundaries for the KM estimator are determined by the event times themselves.

The *hazard function* is estimated as:
$$h(t_{im}) = \frac{d_i}{b_i \left(n_i - \frac{w_i}{2} - \frac{d_i}{2} \right)}$$
 where t_{im} is the midpoint of the interval $[t_{i-1}, t_i)$.

The denominator is an estimate of the total exposure time, T_i , in interval i . For a single individual, exposure time is just the width of the individual, b_i . Censored individuals and those experiencing an event during the interval are assumed to be exposed for only half the interval.

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: The Life-Table Method

Plots of the survival and hazard functions can be obtained with the option `PLOTS(S,H)` in the `PROC LIFETEST` statement.



The sharp drop-off in the last interval is an artifact of how the interval was created.

The length of the last interval is 10, from 50 to 60, while there are no observations after 52.

This leads to an overestimate of the total exposure time in the denominator of the hazard function estimator.

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: The Life-Table Method


We can correct for this by changing the value of week to 53 for all censored observations. Note that there were cases where the event (arrest) occurred in week 52.

```
data survival.recid2;
    set survival.recid;
    if arrest = 0 then week = 53;
run;

title "Lifetable method estimate of the hazard function";
proc lifetest data=survival.recid2 method=life plots=(h)
    intervals=10 20 30 40 50 53;
    time week*arrest(0);
run;
```

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: The Life-Table Method

Now we only see a slight tendency for the hazard function to decline at the end.



Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: Life Tables from Grouped Data

Sometimes time-event data is only available in interval-form instead of specific individual times. Life-tables can still be constructed from published data with the following information:

- the boundaries of the intervals
- the number of events in each interval
- the number of censored cases in each interval

Consider the following survival data for 68 patients from the Stanford Heart Transplantation Program:

| <u>Number of Days</u> | <u>Number of Deaths</u> | <u>Number of Censored</u> |
|-----------------------|-------------------------|---------------------------|
| 0-50 | 16 | 3 |
| 50-100 | 13 | 0 |
| 100-200 | 4 | 2 |
| 200-400 | 5 | 4 |
| 400-700 | 2 | 6 |
| 700-1000 | 4 | 3 |
| 1000-1300 | 1 | 3 |
| 1300-1600 | 1 | 1 |
| 1600+ | 0 | 1 |

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: Life Tables from Grouped Data

```
data stanford;
  input time status number;
  cards;
25 1 16
25 0 3
75 1 11
75 0 0
150 1 4
150 0 2
300 1 5
300 0 4
550 1 2
550 0 6
850 1 4
850 0 3
1150 1 1
1150 0 2
1450 1 1
1450 0 3
1700 1 0
1700 0 1
;
```

Any time within each interval can be entered as the value for time.

Here, the midpoint is used, e.g., 25 for [0, 50)

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: Life Tables from Grouped Data

```

title "Life-tables with grouped data";
proc lifetest data=stanford method=life intervals=50 100 200
              400 700 1000 1300 1600 plots=(s,h);
  time time*status(0);
  freq number;
run;

```

Life Table Survival Estimates

| Interval [Lower, Upper) | Number Failed | Number Censored | Effective Sample Size | Conditional Probability of Failure | Conditional Probability Standard Error | Survival | Failure | Survival Standard Error | |
|----------------------------|------------------|--------------------|-----------------------------|--|---|----------|---------|-------------------------------|--------|
| 0 | 50 | 16 | 3 | 66.5 | 0.2406 | 0.0524 | 1.0000 | 0 | 0 |
| 50 | 100 | 11 | 0 | 49.0 | 0.2245 | 0.0596 | 0.7594 | 0.2406 | 0.0524 |
| 100 | 200 | 4 | 2 | 37.0 | 0.1081 | 0.0510 | 0.5889 | 0.4111 | 0.0608 |
| 200 | 400 | 5 | 4 | 30.0 | 0.1667 | 0.0680 | 0.5253 | 0.4747 | 0.0620 |
| 400 | 700 | 2 | 6 | 20.0 | 0.1000 | 0.0671 | 0.4377 | 0.5623 | 0.0628 |
| 700 | 1000 | 4 | 3 | 13.5 | 0.2963 | 0.1243 | 0.3939 | 0.6061 | 0.0637 |
| 1000 | 1300 | 1 | 2 | 7.0 | 0.1429 | 0.1323 | 0.2772 | 0.7228 | 0.0664 |
| 1300 | 1600 | 1 | 3 | 3.5 | 0.2857 | 0.2415 | 0.2376 | 0.7624 | 0.0677 |
| 1600 | . | 0 | 1 | 0.5 | 0 | 0 | 0.1697 | 0.8303 | 0.0750 |

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: Life Tables from Grouped Data

Evaluated at the Midpoint of the Interval

| Interval [Lower, Upper) | | Median Residual Lifetime | Median Standard Error | PDF | PDF Standard Error | Hazard | Hazard Standard Error |
|----------------------------|------|--------------------------------|-----------------------------|----------|--------------------------|----------|-----------------------------|
| 0 | 50 | 257.7 | 140.1 | 0.00481 | 0.00105 | 0.00547 | 0.001355 |
| 50 | 100 | 686.6 | 139.4 | 0.00341 | 0.000935 | 0.005057 | 0.001513 |
| 100 | 200 | 855.7 | 124.4 | 0.000637 | 0.000308 | 0.001143 | 0.00057 |
| 200 | 400 | 910.5 | 363.2 | 0.000438 | 0.000186 | 0.000909 | 0.000405 |
| 400 | 700 | 982.9 | 216.3 | 0.000146 | 0.000100 | 0.000351 | 0.000248 |
| 700 | 1000 | 779.6 | 236.9 | 0.000389 | 0.000175 | 0.001159 | 0.000571 |
| 1000 | 1300 | . | . | 0.000132 | 0.000126 | 0.000513 | 0.000511 |
| 1300 | 1600 | . | . | 0.000226 | 0.000202 | 0.001111 | 0.001096 |
| 1600 | . | . | . | . | . | . | . |

Summary of the Number of Censored and Uncensored Values

| Total | Failed | Censored | Percent Censored |
|-------|--------|----------|---------------------|
| 68 | 44 | 24 | 35.29 |

NOTE: There were 2 observations with missing values, negative time values or frequency values less than 1.

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: Life Tables from Grouped Data

Note that rapid decline in the hazard function after about 200 days. It is fairly constant after 200 days.

This is reflected in the Median Residual Lifetime column. At time 0, the median residual lifetime of 257.7 days is an estimate of the median survival time for the whole sample.

Of those patients still alive at 50 days, the median residual lifetime rises to 686.6 days.

After 400 days, the median residual lifetime reaches a peak of 982.9 days.

Hazard Function Estimate



Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: Testing for Effects of Covariates

PROC LIFETEST can be used to **test for association** between covariates and survival time.

The tests can be either be **variable by variable** or testing whether a single variable is associated with survival time **given** a set of other variables.

A **drawback** is that the user has no control over the **sequential order** that variables are entered when testing for the association of one variable given a set of variables.

The tests can be interpreted as **nonparametric tests** of the coefficients of the **accelerated failure time model** (presented in Chapter 4).

Similar tests can be carried out with **PROC LIFEREG** and **PROC PHREG**.

In fact, these PROCs are preferable for this task since they can also calculate coefficient estimates, which **LIFETEST** cannot.

The **log-rank chi-square** test reported by **LIFETEST** is identical to the **score statistic** given by **PHREG** for the null hypothesis that all coefficients are zero. 29

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: Testing for Effects of Covariates

We'll illustrate these tests with the `recid` dataset.

```
title "Testing for the effects of covariates";  
proc lifetest data=survival.recid;  
    time week*arrest(0);  
    test fin age race wexp mar paro prio;  
run;
```

Univariate Chi-Squares for the Log-Rank Test

| Variable | Test Statistic | Standard Deviation | Chi-Square | Pr > Chi-Square |
|-------------|-------------------|-----------------------|----------------|--------------------|
| fin | 10.4256 | 5.3361 | 3.8172 | 0.0507 |
| age | 233.2 | 65.6149 | 12.6318 | 0.0004 |
| race | -2.7093 | 3.5791 | 0.5730 | 0.4491 |
| wexp | 16.4141 | 5.2279 | 9.8580 | 0.0017 |
| mar | 7.1773 | 3.6268 | 3.9164 | 0.0478 |
| paro | 2.9471 | 5.1762 | 0.3242 | 0.5691 |
| prio | -108.8 | 28.5013 | 14.5602 | 0.0001 |

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: Testing for Effects of Covariates

Forward Stepwise Sequence of Chi-Squares for the Log-Rank Test

| Variable | DF | Chi-Square | Pr > Chi-Square | Chi-Square Increment | Pr > Increment |
|-------------|----------|----------------|------------------|----------------------|----------------|
| prio | 1 | 14.5602 | 0.0001 | 14.5602 | 0.0001 |
| age | 2 | 25.4905 | <.0001 | 10.9303 | 0.0009 |
| fin | 3 | 28.8871 | <.0001 | 3.3966 | 0.0653 |
| mar | 4 | 31.0920 | <.0001 | 2.2050 | 0.1376 |
| race | 5 | 32.4214 | <.0001 | 1.3294 | 0.2489 |
| wexp | 6 | 33.2800 | <.0001 | 0.8585 | 0.3541 |
| paro | 7 | 33.3828 | <.0001 | 0.1029 | 0.7484 |

From the previous page with the univariate tests, we see that `prio` (a dummy variable for prior arrests) and `age` are the most significant.

The above sequential tests enter variables in their order of significance from the previous univariate tests.

We see that `age` (which had a univariate chi-square value of 12.63) has an incremental p-value of 10.93 once `prio` is already in the model.

`mar`, which had a significant p-value of 0.0478 in the univariate test, has an insignificant incremental p-value of 0.1376 once `prio`, `age`, and `fin` are already in the model.

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: Log Survival and Smoothed Hazard Plots

Recall from page 15, **cumulative hazard function**: $H(t) = \int_0^t h(u) du$

$$\text{and } H(t) = \int_0^t \frac{f(u)}{S(u)} du = -\int_0^t \frac{1}{S(u)} \left\{ \frac{d}{du} S(u) \right\} du = -\ln \{S(t)\}$$

Note that for a constant hazard function, $h(u) = \lambda$, then $H(t) = \lambda t$.


Thus, a plot of $-\ln \{S(t)\}$ versus t should yield a straight line with origin at 0 when the hazard function is constant, i.e., T has an exponential distribution.

The option **PLOTS = (S,LS)** produces the survival and log-survival curves.

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: Log Survival and Smoothed Hazard Plots

```
title "Log-survival plot for the myelomatosis data";  
proc lifetest data=survival.myel plots=(s,ls);  
  time dur*status(0);  
  symbol v=none;  
  
run;
```

The curve appears to be concave down; implying that the cumulative hazard is increasing at a decreasing rate.




Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: Log Survival and Smoothed Hazard Plots

The option **PLOTS = (LLS)** produces the **Log - Log Survival Plot**, i.e., a plot of $\log[-\log(\hat{S}(t))]$ versus $\log(t)$.

If survival times follow a **Weibull** distribution with a hazard function given by $\log h(t) = \alpha + \beta \log(t)$, then the log-log survival plot should be a straight line with a slope of β .

Since this is a small dataset, not too much weight should be given to these plots.

It is shown later how to get similar plots for the log-normal and log-logistic distributions. Also, the log-survival plot can be applied to residuals to assess model fit.



Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: Log Survival and Smoothed Hazard Plots

By **grouping the data** into (arbitrary) time intervals, the **life-table method** can be used to estimate the hazard function.

With **ungrouped data**, estimates of the hazard function can be very **chaotic**.

Kernel smoothing methods can be applied to provide reasonable plots of $h(t)$.

The author provides a **macro** called **SMOOTH** for this purpose.

The macro uses output from **LIFETEST** as input.

```
title "Using the SMOOTH macro";  
proc lifetest data=survival.recid outsurv=a;  
    time week*arrest(0);  
run;
```


```
%include "d:\Charlie\SIGSTAT\Survival Models\smooth.sas";  
%smooth(data=a, time=week, width=8)
```

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: Log Survival and Smoothed Hazard Plots

DATA=A gives the input dataset
TIME=WEEK identifies event times
WIDTH=8 sets bandwidth for the smoothing function.

The default for WIDTH is $1/5^{\text{th}}$ the range of event times.

Note that the graph is truncated at about 8 weeks and 44 weeks to avoid endpoint inaccuracies.



Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST: Conclusion

- **PROC LIFETEST** is useful for **preliminary analysis** of survival data
- It can be used to **test simple hypotheses** about differences in survival across groups
- For experimental studies, **PROC LIFETEST** gives results analogous to **1-way ANOVA**
- **LIFETEST** is not adequate for **two-factor designs** since it can't handle interactions
- The **TEST** statement is useful for **univariate tests** for a large number of covariates
- The **TEST** statement is **limited** for testing significance while controlling for several other variables
- **PROC LIFEREG** and **PROC PHREG** provide **more in-depth options & tests**
- **Heterogeneity** can result in **misleading hazard function estimates** without controlling for the heterogeneity
- **Chapter 8** addresses heterogeneity and **Chapter 4** shows how to reduce its effect by estimating survivor functions using *residuals* from regression models.